

I-13 薄肉断面桁橋のパソコンによるそりねじり解析

桁橋梁設計コンサルタント 正員 林 義 税
北海道大学 工学部 正員 渡 辺 昇

1. まえがき

薄肉断面より成る桁橋にねじりが作用するときには、その断面の重心、せん断中心、曲げ剛性、そりねじり剛性などの断面積をまず計算しておかなければならない。断面形状が複雑であれば、その断面積を求めることは容易でない。そこで、その断面積などをパソコンを用いて計算するためのプログラム用の簡便式について述べる。

2. 断面積などを計算するための簡便式の誘導

2.1 曲げ

図-1において、 yz 軸の任意原点を0とする。一個の矩形断面において、 b が a にくらべて非常に大きい ($\frac{b}{a} > 10$) ものとする。 $F = a \cdot b$ である。

$$y = y_i + S \cdot \cos \beta, \quad Z = Z_i + S \cdot \sin \beta$$

$$b \cdot \cos \beta = y_k - y_i, \quad b \cdot \sin \beta = Z_k - Z_i \quad \text{であるから}$$

$$\begin{aligned} F_y &= \int_F y \cdot dF = \int_i^k (y_i + S \cdot \cos \beta) a \cdot dS = a \left[y_i \cdot S + \frac{\cos \beta}{2} S^2 \right]_i^k \\ &= a \left(y_i \cdot b + \frac{\cos \beta}{2} b^2 \right) = a \cdot b \left(y_i + \frac{1}{2} b \cdot \cos \beta \right) \\ &= F \left\{ y_i + \frac{1}{2} (y_k - y_i) \right\} = \frac{F}{2} (y_i + y_k) \quad \cdots \cdots (1) \end{aligned}$$

$$F_z = \int_F Z \cdot dF = \int_i^k (Z_i + S \cdot \sin \beta) a \cdot dS = \frac{F}{2} (Z_i + Z_k) \quad \cdots \cdots (2)$$

$$\begin{aligned} F_{yz} &= \int_F y \cdot Z \cdot dF = \int_i^k (y_i + S \cdot \cos \beta) (Z_i + S \cdot \sin \beta) a \cdot dS \\ &= a \int_i^k (y_i \cdot Z_i + y_i \cdot \sin \beta \cdot S + Z_i \cdot \cos \beta \cdot S + \sin \beta \cdot \cos \beta \cdot S^2) dS \\ &= a \left[y_i \cdot Z_i \cdot S + (y_i \cdot \sin \beta + Z_i \cdot \cos \beta) \frac{S^2}{2} + \sin \beta \cdot \cos \beta \frac{S^3}{3} \right]_i^k \\ &= a \left\{ y_i \cdot Z_i \cdot b + (y_i \cdot \sin \beta + Z_i \cdot \cos \beta) \frac{b^2}{2} + \sin \beta \cdot \cos \beta \frac{b^3}{3} \right\} \\ &= a \cdot b \left\{ y_i \cdot Z_i + \frac{1}{2} y_i (Z_k - Z_i) + \frac{1}{2} Z_i (y_k - y_i) + \frac{1}{3} (y_k - y_i) (Z_k - Z_i) \right\} \\ &= F \left\{ \frac{1}{3} (y_i \cdot Z_i + y_k \cdot Z_k) + \frac{1}{6} (y_i \cdot Z_k + y_k \cdot Z_i) \right\} \quad \cdots \cdots (3) \end{aligned}$$

$$F_{yy} = \int_F y^2 \cdot dF = \frac{F}{3} (y_i^2 + y_i \cdot y_k + y_k^2) \quad \cdots \cdots (4)$$

$$F_{zz} = \int_F Z^2 \cdot dF = \frac{F}{3} (Z_i^2 + Z_i \cdot Z_k + Z_k^2) \quad \cdots \cdots (5)$$

2.2 そりねじり

図-2において、任意のねじり中心をDとし、ねじりの始点をAとする。 $W(S) = \int_A^S r_t \cdot dS \quad \cdots \cdots (6)$ である。

図-2の場合は、 $w_i = r_{t1} \cdot S_1$, $w_k = w_i + r_{t2} \cdot S_2$ である。

式(1)から式(5)までを応用すれば、一個の矩形断面に対して次式をうる。

$$F_w = \int_F W \cdot dF = \frac{F}{2} (w_i + w_k) \quad \cdots \cdots (7)$$





図-2


$$F_{yW} = \int_F y \cdot W \cdot dF = F \left\{ -\frac{1}{3} (y_i \cdot W_i + y_k \cdot W_k) + \frac{1}{6} (y_i \cdot W_k + y_k \cdot W_i) \right\} \quad \cdots \cdots (8)$$

$$F_{zW} = \int_F z \cdot W \cdot dF = F \left\{ -\frac{1}{3} (z_i \cdot W_i + z_k \cdot W_k) + \frac{1}{6} (z_i \cdot W_k + z_k \cdot W_i) \right\} \quad \cdots \cdots (9)$$

$$F_{WW} = \int_F W^2 \cdot dF = \frac{F}{3} (W_i^2 + W_i \cdot W_k + W_k^2) \quad \cdots \cdots (10)$$

3. y z 軸に関する断面値と $\bar{y} \bar{z}$ 軸に関する断面値との関係

図-3において、y z 軸の任意原点をO、任意のねじり中心をDとする。 $\bar{y} \bar{z}$ 軸の原点は重心Sであり、重心Sのy z 軸に関する座標は、 (y_0, z_0) である。Tはせん断中心であり、そのy z 軸に関する座標は (y_t, z_t) である。



次式の関係がある。

$$\bar{y} = y - y_0 = y - \frac{F_y}{F} \quad \cdots \cdots (11) \quad \bar{z} = z - z_0 = z - \frac{F_z}{F} \quad \cdots \cdots (12) \quad \bar{w} = w - w_0 = w - \frac{F_w}{F} \quad \cdots \cdots (13)$$

$$F_{\bar{y}\bar{z}} = \int_F \bar{y} \cdot \bar{z} \cdot dF = F_{yz} - \frac{F_y \cdot F_z}{F} \quad \cdots \cdots (14) \quad F_{\bar{y}\bar{w}} = \int_F \bar{y} \cdot \bar{w} \cdot dF = F_{yw} - \frac{F_y \cdot F_w}{F} \quad \cdots \cdots (15)$$

$$F_{\bar{z}\bar{w}} = \int_F \bar{z} \cdot \bar{w} \cdot dF = F_{zw} - \frac{F_z \cdot F_w}{F} \quad \cdots \cdots (16) \quad F_{\bar{y}\bar{y}} = \int_F \bar{y}^2 \cdot dF = F_{yy} - \frac{F_y \cdot F_y}{F} \quad \cdots \cdots (17)$$

$$F_{\bar{z}\bar{z}} = \int_F \bar{z}^2 \cdot dF = F_{zz} - \frac{F_z \cdot F_z}{F} \quad \cdots \cdots (18) \quad F_{\bar{w}\bar{w}} = \int_F \bar{w}^2 \cdot dF = F_{ww} - \frac{F_w \cdot F_w}{F} \quad \cdots \cdots (19)$$

$$y_t = \frac{F_{\bar{z}\bar{w}} \cdot F_{\bar{y}\bar{y}} - F_{\bar{y}\bar{w}} \cdot F_{\bar{y}\bar{z}}}{F_{\bar{y}\bar{y}} \cdot F_{\bar{z}\bar{z}} - (F_{\bar{y}\bar{z}})^2} \quad \cdots \cdots (20) \quad z_t = \frac{-F_{\bar{y}\bar{w}} \cdot F_{\bar{z}\bar{z}} + F_{\bar{y}\bar{z}} \cdot F_{\bar{y}\bar{w}}}{F_{\bar{y}\bar{y}} \cdot F_{\bar{z}\bar{z}} - (F_{\bar{y}\bar{z}})^2} \quad \cdots \cdots (21)$$

せん断中心Tをねじりの中心とする場合は次式となる。

$$\bar{w}_T = \bar{w} + z_t \cdot \bar{y} - y_t \cdot \bar{z} \quad \cdots \cdots (22)$$

$$F_{\bar{w}_T \bar{w}_T} = \int_F \bar{w}_T^2 \cdot dF = \int_F (\bar{w} + z_t \cdot \bar{y} - y_t \cdot \bar{z})^2 \cdot dF = F_{\bar{w}\bar{w}} + z_t^2 \cdot F_{\bar{y}\bar{y}} + y_t^2 \cdot F_{\bar{z}\bar{z}} + 2z_t \cdot F_{\bar{y}\bar{w}} - 2y_t \cdot F_{\bar{z}\bar{w}} - 2y_t \cdot z_t \cdot F_{\bar{y}\bar{z}} = F_{\bar{w}\bar{w}} + z_t \cdot F_{\bar{y}\bar{w}} - y_t \cdot F_{\bar{z}\bar{w}} \quad \cdots \cdots (23)$$

$F_{\bar{w}_T \bar{w}_T}$ を一般に C_w で表し、 $E \cdot C_w$ をそりねじり剛性という。(Eはヤング率)

4. r_y, r_z

$$\begin{aligned} \int_F z (y^2 + z^2) \cdot dF &= \int_0^k \{ (z_i + S \cdot \sin \beta) \{ (y_i + S \cdot \cos \beta)^2 + (z_i + S \cdot \sin \beta)^2 \} \} a \cdot dS \\ &= \int_0^k \{ (z_i + S \cdot \sin \beta) \{ (y_i^2 + 2y_i \cdot S \cdot \cos \beta + S^2 \cos^2 \beta) + (z_i^2 + 2z_i \cdot S \cdot \sin \beta + S^2 \sin^2 \beta) \} \} a \cdot dS \\ &= a \int_0^k \{ (y_i^2 \cdot z_i + z_i^3) + S(2y_i \cdot z_i \cdot \cos \beta + 2z_i^2 \cdot \sin \beta + y_i^2 \cdot \sin \beta + z_i^2 \cdot \sin \beta) \\ &\quad + S^2(z_i \cdot \cos^2 \beta + z_i \cdot \sin^2 \beta + 2y_i \cdot \sin \beta \cdot \cos \beta + 2z_i \cdot \sin^2 \beta) + S^3(\sin \beta \cdot \cos^2 \beta \\ &\quad + \sin \beta \cdot \sin^2 \beta) \} a \cdot dS = a \left[b(y_i^2 \cdot z_i + z_i^3) + \frac{b^2}{2} \{ 2y_i \cdot z_i \cdot \frac{1}{b} (y_k - y_i) + 2z_i^2 \cdot \frac{1}{b} (z_k - z_i) \right. \\ &\quad + y_i^2 \cdot \frac{1}{b} (z_k - z_i) + z_i^2 \cdot \frac{1}{b} (z_k - z_i) \} + \frac{b^3}{3} \{ z_i \cdot \frac{1}{b^2} (y_k - y_i)^2 + z_i \cdot \frac{1}{b^2} (z_k - z_i)^2 \\ &\quad + 2y_i \cdot \frac{1}{b^2} (y_k - y_i)(z_k - z_i) + 2z_i \cdot \frac{1}{b^2} (z_k - z_i)^2 \} + \frac{b^4}{4} \{ \frac{1}{b^3} (z_k - z_i)(y_k - y_i)^2 \\ &\quad + \frac{1}{b^3} (z_k - z_i)^3 \} \} = \frac{a \cdot b}{12} \{ (z_i + z_k)(3z_i^2 + 2y_i \cdot y_k + 3z_k^2) + y_i^2(3z_i + z_k) + y_k^2(z_i + 3z_k) \\ &\quad + y_i^2(z_i + 3z_k) \} \end{aligned}$$

$$r_y = \frac{1}{F_{zz}} \int_F z (y^2 + z^2) \cdot dF = \frac{F}{12 F_{zz}} \{ (z_i + z_k)(3z_i^2 + 2y_i \cdot y_k + 3z_k^2) + y_i^2(3z_i + z_k) + y_k^2(z_i + 3z_k) \} \quad \cdots \cdots (24)$$

$$r_z = \frac{1}{F_{yy}} \int_F y (y^2 + z^2) \cdot dF = \frac{F}{12 F_{yy}} \{ (y_i + y_k)(3y_i^2 + 2z_i \cdot z_k + 3y_k^2) + z_i^2(3y_i + y_k) + z_k^2(y_i + 3y_k) \} \quad \cdots \cdots (25)$$

5. パソコンのプログラム

```

10 REM-WAPPING-TORSION---
20 READ M
30 DIM T(M),YI(M),ZI(M),YK(M),ZK(M),B(M),F(M)
40 DIM WI(M),WK(M),WTI(M),WTK(M)
50 FOR I=1 TO M
60 READ T(I),YI(I),ZI(I),YK(I),ZK(I)
70 NEXT I
80 FOR I=1 TO M
90 READ WI(I),WK(I)
100 NEXT I
110 FOR I=1 TO M
120 B(I)=SQR((YK(I)-YI(I))^2+(ZK(I)-ZI(I))^2)
130 F(I)=T(I)*B(I)
140 NEXT I
150 F=0:FY=0:FZ=0:FW=0:FYZ=0:FYY=0:FZZ=0:FYW=0:FZW=0:FWW=0
160 FOR I=1 TO M
170 F=F+F(I)
180 FY=FY+(YI(I)+YK(I))*F(I)/2
190 FZ=FZ+(ZI(I)+ZK(I))*F(I)/2
200 FW=FW+(WI(I)+WK(I))*F(I)/2
210 FYZ=FYZ+((YI(I)*ZI(I)+YK(I)*ZK(I))/3+(YI(I)*ZK(I)+YK(I)*ZI(I))/6)*F(I)
220 FYY=FYY+(YI(I)^2+YI(I)*YK(I)+YK(I)^2)*F(I)/3
230 FZZ=FZZ+(ZI(I)^2+ZI(I)*ZK(I)+ZK(I)^2)*F(I)/3
240 FYW=FYW+((YI(I)*WI(I)+YK(I)*WK(I))/3+(YI(I)*WK(I)+YK(I)*WI(I))/6)*F(I)
250 FZW=FZW+((ZI(I)*WI(I)+ZK(I)*WK(I))/3+(ZI(I)*WK(I)+WK(I)*WI(I))/6)*F(I)
260 FWW=FWW+(WI(I)^2+WI(I)*WK(I)+WK(I)^2)*F(I)/3
270 NEXT I
280 FOR I=1 TO M
290 LPRINT USING"###.#####^";T(I),YI(I),ZI(I),YK(I),ZK(I),B(I),F(I):LPRINT
300 LPRINT USING"###.#####^";WI(I),WK(I):LPRINT
310 NEXT I
320 LPRINT
330 LPRINT USING"###.#####^";F,FY,FZ,FYZ,FYY,FZZ:LPRINT
340 LPRINT USING"###.#####^";FW,FYW,FZW,FWW:LPRINT
350 LPRINT
360 YO=FY/F:ZO=FZ/F:WO=FW/F
370 FYZ=FYZ-FY*FZ/F:FYY=FYY-FY^2/F:FZZ=FZZ-FZ^2/F
380 FYW=FYW-FY*FW/F:FZW=FZW-FZ*FW/F:FWW=FWW-FW^2/F
390 LPRINT USING"###.#####^";YO,ZO,FYZ,FYY,FZZ:LPRINT
400 LPRINT USING"###.#####^";WO,FYW,FZW,FWW:LPRINT
410 LPRINT :LPRINT
420 YT=(FZW*FYY-FYW*FYZ)/(FYY*FZZ-FYZ^2)
430 ZT=(-FYW*FZZ+FZW*FYZ)/(FYY*FZZ-FYZ^2)
440 FOR I=1 TO M
450 WTI(I)=(WI(I)-WO)+ZT*(YI(I)-YO)-YT*(ZI(I)-ZO)
460 WTK(I)=(WK(I)-WO)+ZT*(YK(I)-YO)-YT*(ZK(I)-ZO)
470 NEXT I
480 FOR I=1 TO M
490 LPRINT USING"###.#####^";WTI(I),WTK(I):LPRINT
500 NEXT I
510 FWWT1=0
520 FOR I=1 TO M
530 FWWT1=FWWT1+(WTI(I)^2+WTI(I)*WTK(I)+WTK(I)^2)*F(I)/3
540 NEXT I
550 FWWT2=FWWT1+ZT*FYW-YT*FZW
560 LPRINT USING"###.#####^";YT,ZT,FWWT1,FWWT2:LPRINT
570 LPRINT :LPRINT
580 FOR I=1 TO M
590 YSI(I)=YI(I)-YO:YSK(I)=YK(I)-YO:ZSI(I)=ZI(I)-ZO:ZSK(I)=ZK(I)-ZO
600 RZ(I)=((YSI(I)+YSK(I))*(3*YSI(I)^2+2*ZSI(I)*ZSK(I)+3*YSK(I)^2)+ZSI(I)^2
*(3*YSI(I)+YSK(I))+ZSK(I)^2*(YSI(I)+3*YSK(I)))*F(I)/12
610 RY(I)=((ZSI(I)+ZSK(I))*(3*ZSI(I)^2+2*YSI(I)*YSK(I)+3*ZSK(I)^2)+YSI(I)^2
*(3*ZSI(I)+ZSK(I))+YSK(I)^2*(ZSI(I)+3*ZSK(I)))*F(I)/12
620 NEXT I
630 FOR I=1 TO M
640 LPRINT USING"###.#####^";YSI(I),ZSI(I),YSK(I),ZSK(I):LPRINT
650 NEXT I
660 LPRINT :LPRINT
670 RY=0:RZ=0
680 FOR I=1 TO M
690 RY=RY+RY(I):RZ=RZ+RZ(I)
700 NEXT I
710 RY=RY/FZZ:RZ=RZ/FYY
720 LPRINT USING"###.#####^";RY,RZ:LPRINT
730 LPRINT :LPRINT
740 END

```

6. 解析例（そりねじり横倒れ座屈）




図-4 横断面図

図-4のような逆π形の断面をもつプレートガーダーに、図-6のような床版コンクリートを現場で打設する。

そのときの横倒れ座屈（そりねじり

座屈）を解析してみよう。

前頁のパソコンプログラムに、図-4

と図-5により、右記のようなデータ

を入力すると、次の結果をうる。

重心の座標は（157.5cm，95.6cm），

せん断中心の座標は（157.5 cm，-86.5 cm）， $r_y=106.6$ cm， $r_z=0$

$$F_{yz}=0, F_{yy}=1.54 \times 10^7 \text{ cm}^4, F_{zz}=1.19 \times 10^7 \text{ cm}^4, F_{yz}=1.33 \times 10^9 \text{ cm}^5$$

$$F_{zz}=1.88 \times 10^9 \text{ cm}^5, F_{yy}=5.50 \times 10^{11} \text{ cm}^6, C_w = F_{yz} + F_{yz} = 1.38 \times 10^{11} \text{ cm}^6, J_T = 2053 \text{ cm}^4$$

図-6の場合の横倒れ座屈荷重 P_{cr} を次に計算してみよう。

鋼断面の重心Sとコンクリート断面の重心Cとの間の距離を $e=174.2$ cm とする。

$$P_{cr} = \frac{9.2\pi^2}{l^4} E \cdot F_{yy} \left[(Z_T - 0.466e - 0.267 r_y)^2 + \sqrt{(Z_T - 0.466e - 0.267 r_y)^2 + \frac{l^2 G \cdot J_T}{\pi^2 E \cdot F_{yy}} + \frac{C_w}{F_{yy}}} \right] \\ = \frac{9.2 \times 3.14^2}{5080^4} \times 2.1 \times 10^6 \times 1.54 \times 10^7 \left[(-182.1 - 0.466 \times 174.2 - 0.267 \times 106.6)^2 + \sqrt{(-182.1 - 0.466 \times 174.2 - 0.267 \times 106.6)^2 + \frac{5080^2 \times 0.8 \times 10^6 \times 2053}{3.14^2 \times 2.1 \times 10^6 \times 1.54 \times 10^7} + \frac{1.38 \times 10^{11}}{1.54 \times 10^7}} \right] = 66.8 \text{ kg/cm}$$

安全率 2.5 で除すと、 $P_a = \frac{66.8}{2.5} = 26.72 \text{ kg/cm} = 2672 \text{ kg/m}$

結局、 $P_a=2672 \text{ kg/m}$ 以上の重さの床版コンクリートを打設すると、この鋼桁は横倒れ座屈を起こして崩壊する。




図-5 W(s)図




図-6